
Abstract

The worker/wrapper transformation is a general technique for improving the performance of re-
cursive programs by changing their types. The previous formalisation (Gill & Hutton, 2009) was
based upon a simple fixed point semantics of recursion. In this article we develop a more structured
approach, based upon initial algebra semantics. In particular, we show how the worker/wrapper
transformation can be applied to programs defined using the structured pattern of recursion captured
by fold operators, and illustrate our new technique with a number of examples.

1 Introduction

The worker/wrapper transformation is a general technique for changing the type of a
recursive program to improve its performance. The basic idea is simple and pervasive:
given a recursive program of some type, we aim to factorise it into a more efficient worker
program of a different type, together with a wrapper program that acts as an interface
between the original program and the new worker.

Special cases of the worker/wrapper transformation have been utilised for many years,
particularly in optimizing compilers. For example, the technique has been used in the
Glasgow Haskell Compiler since its inception, to replace the use of boxed data structures
by more efficient unboxed data structures when safe to do so (Peyton Jones & Launch-
bury, 1991). However, it is only recently that the technique has been formalised, proved
correct, and presented as a general approach to improving the performance of programs by
improving the choice of data structures (Gill & Hutton, 2009).

The previous formalisation of the technique was based upon a simple fixed point seman-
tics of recursive programs. In this article we develop a more structured approach, based
upon initial algebra semantics. In particular, we show how the worker/wrapper transforma-
tion can be applied to programs defined using the structured pattern of recursion captured
by fold operators. More precisely, the article makes the following contributions:

• We show how the worker/wrapper transformation applies to programs defined using
folds, by generalising to a categorical view of types as initial algebras.

• We identify four conditions for the correctness of the transformation, and show that
these conditions form a simple lattice structure.

• We illustrate our technique with a number of examples, including a correctness proof
for a new approach to implementing substitution efficiently (Voigtländer, 2008).

The article is aimed at readers who are familiar with the basics of initial algebra seman-
tics, say to the level of chapter two of Bird & de Moor (1997), but no previous experience
with the worker/wrapper transformation is assumed. An extended version of the article that
includes all the proofs is available from the authors’ web pages.

2 Initial algebra semantics

The recursion operator fold encapsulates a common pattern for defining functions that
consume values of a recursively defined type (Hutton, 1999). In this section we review the
categorical treatment of fold in terms of initial algebras, and introduce our notation. For
further details, see (Malcolm, 1990; Meijer et al., 1991; Bird & de Moor, 1997).

Suppose that we fix a category C and a functor F : C → C on this category. Then the
notion of an algebra is defined as a pair (A, f ) comprising an object A and an arrow f :
FA → A, and a homomorphism h : (A, f ) → (B,g) from one such algebra to another is an
arrow h : A → B such that the following diagram commutes:

FA
Fh ��

f

��

FB

g

��
A

h
�� B

Algebras and homomorphisms themselves form a category, with composition and identities
inherited from C. An initial algebra is an initial object in this new category, and we write
(µF, in) for an initial algebra, and fold f for the unique homomorphism h : (µF, in) →
(A, f ) from the initial algebra to any other algebra (A, f ). That is, fold f is defined as the
unique arrow that makes the following diagram commute:

FµF
F (fold f ) ��

in

��

FA

f

��
µF

fold f
�� A

In the literature, fold f is sometimes written using the banana brackets notation (| f |), and
termed a catamorphism. The above definition for fold f can also be expressed as the
following equivalence, known as the universal property of fold:

h = fold f ⇔ h ◦ in = f ◦ Fh

The ⇒ direction states that fold f is a homomorphism from the initial algebra (µF, in) to
another algebra (A, f ), while the ⇐ direction states that any other such homomorphism h

must be equal to fold f . Taken as a whole, the universal property expresses in an equational
manner the fact that fold f is the unique homomorphism from (µF, in) to (A, f ).

The universal property can be used to verify the well-known fusion property of fold,
which states that the composition of a function and a fold can always be re-expressed as a
single fold, provided the function is a homomorphism of the appropriate type:

h ◦ f = g ◦ Fh

h ◦ fold f = fold g

Proof :

h ◦ fold f = fold g
⇔ { universal property of fold }

h ◦ fold f ◦ in = g ◦ F (h ◦ fold f )
⇔ { F is a functor }

h ◦ fold f ◦ in = g ◦ Fh ◦ F (fold f )
⇔ { fold f is a homomorphism }

h ◦ f ◦ F (fold f ) = g ◦ Fh ◦ F (fold f )
⇐ { extensionality }

h ◦ f = g ◦ Fh
��

2.1 Example: finite lists

Suppose that we define a functor L on the category SET by L A = 1 +(Z×A), where Z is
the integers. Then an algebra is a pair (A, f ) comprising a set A and a function f : 1+(Z×
A)→ A. Functions of this type can always be uniquely decomposed into the form f = [g,h]
for some g : 1 → A and h : Z×A → A, and a homomorphism k : (A, [g,h])→ (B, [i, j]) from
one such algebra to another is given by a function k : A → B such that

k ◦ g = i
k ◦ h = j ◦ (id

Z
× k)

The functor L has an initial algebra (µL, in) = (List(Z), [nil,cons]), where List(Z) is the
set of finite lists of integers, and nil : 1 → List(Z) and cons : Z× List(Z) → List(Z) are
constructors for this set. Given any other set A and functions i : 1 → A and j : Z×A → A,
the function fold [i, j] : List(Z) → A is uniquely defined by:

fold [i, j] ◦ nil = i
fold [i, j] ◦ cons = j ◦ (id

Z
× fold [i, j])

That is, fold [i, j] replaces the nil constructor at the end of a list by the function i, and
each cons constructor within the list by the function j. For example, the function sum :
List(Z) → Z that sums a list of integers can be defined by sum = fold [zero, plus], where
zero : 1 → Z and plus : Z×Z → Z are given by zero () = 0 and plus (x,y) = x + y.

3 Worker/wrapper for folds

Consider the problem of changing the return type of a fold to improve its performance.
More precisely, suppose we are given a function fold f : µF → A for some f : FA → A,
and we wish to change the return type from A to some other type B. The worker/wrapper
approach to this problem is based upon the use of conversion functions

A

rep
��
B

abs

��

with the property that abs ◦ rep = idA. This equation states that converting a value of the
original type into the new type and then back again does not change the value, and in the
terminology of data representation (Hoare, 1972) expresses that the abstract type A can be
faithfully represented by the concrete type B. Given such a setting, we now seek conditions
under which the following diagram commutes:

µF

fold f

����
��

��
��

��
��

�
fold g

���
��

��
��

��
��

��

A B
abs

��

That is, in worker/wrapper terminology, we seek conditions that allow the original recur-
sive function fold f that produces a result of type A to be factorised as the composition of
a recursive worker function fold g that produces a result of type B, and a wrapper function
abs that converts the result back to the original type A.

One approach to solving this problem is to simply apply fusion. Even though this prop-
erty is normally viewed as being concerned with combining a function with a fold, it
can also be viewed in the opposite direction as providing a sufficient condition for the
factorisation, or fission (Gibbons, 2006), of a fold:

fold f = abs ◦ fold g
⇐ { fusion property of fold }

f ◦ F abs = abs ◦ g

However, given the assumption that abs ◦ rep = idA, we can in fact identify four relevant
conditions, of which the equation arising from fusion is just one instance:

(1) g = rep ◦ f ◦ F abs

(2) rep ◦ f = g ◦ F rep

(3) f ◦ F abs = abs ◦ g

(4) f = abs ◦ g ◦ F rep

These equations correspond to the four possible ways of completing the following diagram
by replacing each question mark with either rep or abs:

FA

f

��

F?
FB

g

��
A

?
B

What do these conditions express, and how do they relate? Equation (1) provides an
explicit definition for g in terms of f ; (2) states that rep is a homomorphism from f to g;
(3) states that abs is a homomorphism from g to f (the condition that directly arises from
the use of fusion); and (4) provides a definition for f in terms of g. Together, they form a
simple lattice, with (1) as the strongest condition and (4) as the weakest:

(4)

(2)

������
����

(3)

		����
����

(1)

		����
����

������
����

For example, (1) ⇒ (2) can be verified as follows:

g ◦ F rep = rep ◦ f
⇔ { identities }

g ◦ F rep = rep ◦ f ◦ idFA
⇔ { F is a functor }

g ◦ F rep = rep ◦ f ◦ F idA
⇔ { abs ◦ rep = idA }

g ◦ F rep = rep ◦ f ◦ F (abs ◦ rep)
⇔ { F is a functor }

g ◦ F rep = rep ◦ f ◦ F abs ◦ F rep
⇐ { extensionality }

g = rep ◦ f ◦ F abs
��

It is now straightforward to verify that each of the first three conditions implies the
desired factorisation result, namely that fold f = abs ◦ fold g. In particular, we already
know that (1) implies (3) using the lattice diagram, and that (3) implies the desired result
using fusion, hence it only remains to verify that condition (2) is also sufficient:

fold f = abs ◦ fold g
⇔ { identities }

idA ◦ fold f = abs ◦ fold g
⇔ { abs ◦ rep = idA }

abs ◦ rep ◦ fold f = abs ◦ fold g
⇐ { extensionality }

rep ◦ fold f = fold g
⇐ { fusion property of fold }

rep ◦ f = g ◦ F rep
��

The situation regarding (4) is more complicated, and we will return to this shortly. In the
meantime, let us consider how the first three conditions are used in practice.

For some applications, the definition for the function g that forms the body of the worker
fold g will already be given, and our aim then is to verify that one of the three conditions
is satisfied, to ensure that the worker/wrapper factorisation holds. For many applications,
however, our aim will be to construct a suitable function g. In such cases, condition (1)
provides an explicit definition g = rep ◦ f ◦ F abs for the body of the worker in a similar
manner to (Gill & Hutton, 2009), and our aim then is to simplify the definition. This
simplification process is typically driven by the desire to fuse together instances of rep and
abs, to eliminate the overhead of repeatedly converting between the concrete and abstract
types. In contrast, conditions (2) and (3) provide a specification for g, and our aim is then to
calculate a definition that satisfies the specification, again with the desire to fuse together
instances of the conversion functions between the two types.

Given that (1) is the strongest condition and provides an explicit definition for g as a
starting point, why would we ever wish to use the other conditions? In our experience,
using one of the weaker conditions often results in a simpler verification or calculation
process. In combination with the fact that (3) corresponds to the familiar case of fusion, for
the remainder of the article we will primarily focus on (2). Nonetheless, it is interesting to
consider the other conditions, and their relationships.

3.1 The weakest condition

Let us now return to the remaining condition in our lattice:

(4) f = abs ◦ g ◦ F rep

Unfortunately, in general this condition does not imply that fold f = abs ◦ fold g, and is
only sufficient to ensure the following more specialised worker/wrapper factorisation in
which the body g of the worker is composed with an additional term:

fold f = abs ◦ fold (g ◦ F (rep ◦ abs))

Proof :

fold f = abs ◦ fold (g ◦ F (rep ◦ abs))
⇐ { fusion property of fold }

abs ◦ g ◦ F (rep ◦ abs) = f ◦ F abs
⇔ { F is a functor }

abs ◦ g ◦ F rep ◦ F abs = f ◦ F abs
⇐ { extensionality }

abs ◦ g ◦ F rep = f

��
The additional term F (rep ◦ abs) in the worker plays the role of a normalisation function
that is applied after each recursive call. In general, rep ◦ abs �= idB, but we can think of
rep ◦ abs as normalising a value of type B by first converting to the type A, which is
typically a ‘smaller’ type, and then converting back to B.

It is natural to ask when (4) does imply fold f = abs ◦ fold g. The answer is given by the
following condition, which states that rep ◦ abs is a homomorphism from g to itself:

(5) rep ◦ abs ◦ g = g ◦ F (rep ◦ abs)

In particular, we then have the following equivalence:

(4)∧ (5) ⇔ (2)∧ (3)

That is, the combination of (4) and (5) is equivalent to the combination of (2) and (3), either
condition of which we have already shown implies the worker/wrapper factorisation.

We conclude this section by noting that condition (5) also implies the following property,
which is precisely the worker/wrapper fusion property from (Gill & Hutton, 2009) for the
special case when the worker is defined using fold:

(6) rep ◦ abs ◦ fold g = fold g

That is, even though the identity rep ◦ abs = idB does not always hold, given (5) this
identity does hold for the special case of values of type B that are produced by the worker
itself. The proof of worker/wrapper fusion is now a simple application of fusion:

rep ◦ abs ◦ fold g = fold g
⇐ { fusion property of fold }

rep ◦ abs ◦ g = g ◦ F (rep ◦ abs)
��

4 Worker/wrapper for lists

To illustrate our new worker/wrapper technique, we now move from the abstract world of
category theory to the concrete world of Haskell (Peyton Jones, 2003). Our first example
concerns lists, for which the fold operator in Haskell is defined as follows:

fold :: (a → b → b) → b → [a ] → b
fold f v [ ] = v
fold f v (x : xs) = f x (fold f v xs)

That is, the function fold f v processes a list by replacing the empty list [ ] by the value v,
and each constructor (:) within the list by the function f . For example, the function that
sums a list of numbers can be defined by sum = fold (+) 0.

Now suppose we are given a function fold f v :: [a ] → b for some f :: a → b → b and
v :: b, and that we wish to change the return type of the fold from b to some other type c.
Moreover, we also assume that we are given conversion functions rep ::b→ c and abs::c→
b satisfying the equation abs ◦ rep = idb. Then instantiating our general theory from the
previous section, we find that any of the three conditions

(1) g x y = rep (f x (abs y))

(2) rep (f x y) = g x (rep y)

(3) f x (abs y) = abs (g x y)

is sufficient to justify the following factorisation of the original fold that produces a result
of type b into the composition of a worker fold that produces a result of type c, and a
wrapper function that converts the result back to the original type b:

fold f v = abs◦ fold g (rep v)

4.1 Example: fast reverse

Consider the problem of transforming a simple function that reverses a list into a more
efficient version that uses accumulation. This transformation is normally achieved using
more elementary techniques (Hutton, 2007), but we now show that it also fits naturally
into our worker/wrapper paradigm based upon fold, and leads to a simpler derivation than
the previous worker/wrapper approach based upon fix.

Using explicit recursion, a reverse function can be defined by

rev :: [a ] → [a ]
rev [ ] = [ ]
rev (x : xs) = rev xs++[x ]

or equivalently, using the fold operator for lists:

rev :: [a ] → [a ]
rev = fold snoc [ ]

snoc :: a → [a ] → [a ]
snoc x xs = xs++[x ]

However, because of the use of append (++), this definition for rev takes quadratic time.
We now show how our worker/wrapper technique for fold can be used to derive a more
efficient worker that uses an extra argument to accumulate the result, together with a
wrapper that takes care of the initial setup. Using the notion of currying, the introduction
of an accumulator argument corresponds to changing the return type of rev from a list to
a function on lists, i.e. changing from the original return type [a ] to the new return type
[a ] → [a ]. The necessary conversion functions between the two types, the latter of which
is sometimes called Hughes lists (Hughes, 1986), are defined as follows:

type H a = [a ] → [a ]

rep :: [a ] → H a
rep xs = (xs ++)

abs :: H a → [a ]
abs h = h [ ]

Note that rep is just a synonym for (++). It is straightforward to verify the worker/wrapper
assumption abs◦ rep = id[a]. We also have the important property that rep forms a monoid
homomorphism from lists to Hughes lists, in the sense that:

rep (xs++ ys) = rep xs◦ rep ys

rep [ ] = id[a]

In the case of reverse, it turns out that the most convenient condition to use as the basis
for constructing the worker function is condition (2):

rep (snoc x xs) = g x (rep xs)

We calculate a function g satisfying this equation as follows:

rep (snoc x xs)
= { applying snoc }

rep (xs++[x ])
= { rep is a homomorphism }

rep xs◦ rep [x ]
= { applying rep }

rep xs◦ (x:)
= { define g x h = h ◦ (x:) }

g x (rep xs)

Now that we have satisfied the necessary preconditions, applying the worker/wrapper
transformation for fold gives the following new definitions:

rev :: [a ] → [a ]
rev = abs◦work

work :: [a ] → H a
work = fold g (rep [ ])

Finally, if we make the list arguments explicit, and then expand out the component func-
tions, we obtain the expected linear time version of reverse that uses an accumulator:

rev :: [a ]→ [a ]
rev xs = work xs [ ]

work :: [a ]→ [a ] → [a ]
work [ ] ys = ys
work (x : xs) ys = work xs (x : ys)

We conclude with a number of observations about the above derivation. First of all, in
common with the previous derivation of fast reverse using the worker/wrapper technique
for fix (Gill & Hutton, 2009), once we have made the decision to use Hughes’ representa-
tion of lists, the rest of the derivation proceeds using simple equational reasoning, without
the need for induction. However, in contrast to the previous derivation, using the additional
structure afforded by using fold avoids the need for the additional functions wrap and
unwrap, the use of worker/wrapper fusion, and the need to expand out the worker as an
essential step in the derivation, resulting in a simpler derivation.

4.2 Example: fast reverse revisited

It is interesting now to return to our earlier question of why we don’t always use condi-
tion (1), which provides an explicit definition for g as a starting point. In the case of the
reverse example, the initial definition would then be as follows:

g x y = rep (snoc x (abs y))

The problem comes when we try and simplify this definition:

g x y
= { applying g }

rep (snoc x (abs y))
= { applying snoc }

rep (abs y ++[x ])
= { rep is a homomorphism }

rep (abs y)◦ rep [x ]
= { applying rep }

rep (abs y)◦ (x:)

Now we appear to be stuck. We’d like to fuse together rep and abs in the final expression to
give the definition g x y = y◦ (x:), but unfortunately it is not the case that rep◦abs = idH a.
In order to make progress, we begin by rewriting the worker

work = fold g (rep [ ])

by making the first list argument explicit, expanding out the fold, and using the above
simplification of g to give the following definition using explicit recursion:

work [ ] = rep [ ]
work (x : xs) = rep (abs (work xs))◦ (x:)

While rep ◦ abs = idH a is not true in general, for the special case of values produced by
worker itself we do have rep ◦ abs◦work = work, the worker/wrapper fusion property (6)
from section 3.1, which allows use to rewrite the worker as

work [ ] = rep [ ]
work (x : xs) = work xs◦ (x:)

which can then be expanded to give the expected definition:

work [ ] ys = ys
work (x : xs) ys = work xs (y : ys)

However, an unsatisfactory aspect of the above derivation is the need to rewrite the
worker using explicit recursion in order to make progress by applying worker/wrapper
fusion. Can the derivation also be performed at the fold level, without expanding out the
recursion? The key to achieving this is to observe that in this context the second argument
of g will always be of the form rep z for some list z, since both the base and recursive
case for the worker have an application of rep at the outer level. Using this assumption, the
definition for g can then be simplified as follows:

g x y
= { previous simplification }

rep (abs y)◦ (x:)
= { assuming y = rep z }

rep (abs (rep z))◦ (x:)
= { abs◦ rep = id }

rep z◦ (x:)
= { assuming y = rep z }

y◦ (x:)

Avoiding the need for this kind of ad-hoc additional reasoning is precisely the benefit
that we obtain by starting from condition (2) rather than (1). In particular, using rep (f x y)=
g x (rep y) as our specification for g makes explicit from the outset that we can assume the
second argument to g is always of the form rep y.

5 Worker/wrapper for expressions

For our next example we move from the type of lists to a simple language of expressions
comprising integers and addition, together with its associated fold operator:

data Expr = Val Int | Add Expr Expr

fold :: (a → a → a) → (Int → a) → Expr → a
fold f v (Val n) = v n
fold f v (Add x y) = f (fold f v x) (fold f v y)

Suppose now that we wish to change the return type of a function fold f v :: Expr → a
from the original type a to some other type b, and that we are given conversion functions
rep :: a → b and abs :: b → a such that abs ◦ rep = ida. In this context, our general theory
from section 3 states that any of the three conditions

(1) g x y = rep (f (abs x) (abs y))

(2) rep (f x y) = g (rep x) (rep y)

(3) f (abs x) (abs y) = abs (g x y)

is sufficient to justify the following factorisation of the original fold that produces a result
of type a into the composition of a worker fold that produces a result of type b, and a
wrapper function that converts the result back to the original type a:

fold f v = abs◦ fold g (rep◦ v)

5.1 Example: continuation-passing evaluation

Consider the problem of transforming an evaluator for expressions into continuation-passing
style, the typical first step in deriving an efficient abstract machine (Hutton & Wright,

2006). Using explicit recursion, an evaluation function can be defined by

eval :: Expr → Int
eval (Val n) = n
eval (Add x y) = eval x + eval y

or equivalently, using the fold operator for expressions:

eval :: Expr → Int
eval = fold (+) id

Rewriting this definition in continuation-passing style involves taking a function on inte-
gers (the continuation) as an extra argument, which using currying corresponds to changing
from the original return type Int to the new return type (Int → Int) → Int. The necessary
conversion functions between the two types are defined as follows:

type Cint = (Int → Int) → Int

rep :: Int → Cint
rep n = λ c → c n

abs :: Cint → Int
abs f = f id

It is easy to show that abs◦ rep = idInt. As with fast reverse, the appropriate starting point
for constructing the worker in this case is condition (2),

rep (x +y) = g (rep x) (rep y)

from which we calculate a function g satisfying this equation as follows:

rep (x +y) c
= { applying rep }

c (x + y)
= { abstracting over x }

(λ n → c (n + y)) x
= { unapplying rep }

rep x (λ n → c (n +y))
= { abstracting over y }

rep x (λ n → (λ m → c (n +m)) y)
= { unapplying rep }

rep x (λ n → rep y (λ m → c (n +m)))
= { define g a b = a (λ n → b (λ m → c (n +m))) }

g (rep x) (rep y)

Now that we have satisfied the necessary preconditions, applying the worker/wrapper
transformation for fold gives the following definitions

eval :: Expr → Int
eval = abs◦work

work :: Expr → Cint
work = fold g (rep◦ id)

which expand out to give the expected continuation-passing evaluator:

eval :: Expr → Int
eval e = work e id

work :: Expr → (Int → Int) → Int
work (Val n) c = c n
work (Add x y) c = work x (λ n → work y (λ m → c (n +m)))

Once again, note that the derivation proceeds using simple equational reasoning and does
not require induction. Moreover, in contrast to our previous derivation of such an evaluator
using more elementary techniques (Hutton & Wright, 2006), using worker/wrapper con-
dition (2) as the starting point results in derivation whose goal is made explicit from the
outset, namely to construct a function g such that rep (x+y) = g (rep x) (rep y), rather than
this property being implicit in the structure of the derivation itself.

6 Efficient substitution

For our final example, we consider a more challenging problem: improving the perfor-
mance of monadic substitution on trees. The example is take from (Voigtländer, 2008), but
whereas the author only sketches a proof of correctness and conjectures that a formal proof
may require sophisticated techniques, we show that a simple proof is possible using our
worker/wrapper technique for fold. We begin by defining the type Tree a of binary trees
whose leaves contain values of some parameter type a:

data Tree a = Leaf a | Node (Tree a) (Tree a)

Now recall that in Haskell, the categorical notion of a monad is captured by the following
class declaration, which states that a parameterised type m is a member of the class Monad
of monadic types if it is equipped with return and >>= functions of the specified types:

class Monad m where
return :: a → m a
(>>=) :: m a → (a → m b) → m b

The two functions must also satisfy identity and associativity properties:

return x >>= f = f x

e >>=return = e

(e >>= f )>>=g = e >>=(λ x → f x >>=g)

It is straightforward to make Tree into a monadic type by the following instance decla-
ration, and to verify that the required monad laws are satisfied:

instance Monad Tree where
return :: a → Tree a
return x = Leaf x

(>>=) :: Tree a → (a → Tree b) → Tree b
(Leaf x)>>= f = f x
(Node l r)>>= f = Node (l>>= f ) (r >>= f )

This declaration implements the well-known idea that substitution is monadic. In particu-
lar, if we view values of type Tree a as terms with variables of type a, then return converts
a value into the corresponding term, and t >>= f is the term that results from applying the
substitution f to every variable in the term t. For example, given the tree of characters

t = Node (Leaf ’a’) (Leaf ’b’)

and the substitution

f :: Char → Tree Int
f ’a’ = Leaf 1
f ’b’ = Node (Leaf 2) (Leaf 3)

then the expression t >>= f produces the following tree of integers:

Node (Leaf 1) (Node (Leaf 2) (Leaf 3))

Now consider the following recursive function on natural numbers, which uses substitu-
tion to produce a tree of integers of a specified depth:

fullTree :: Int → Tree Int
fullTree 1 = return 1
fullTree (n +1) = fullTree n >>=λ i →

Node (return (n− i)) (return (i+1))

That is, a tree of depth 1 is produced by returning a leaf, and a tree of depth n + 1 by
recursively building a tree of depth n, and then using substitution to replace each leaf value
i by a tree of depth two with leaf values n− i and i + 1. For example, the first four trees
produced by applying fullTree can be pictured as follows:

1 3 2 2 3 1 0 4

As we would expect from these examples, fullTree takes exponential time. Now consider
the function zigzag that follows a path down a tree that alternates between moving left (zig)
and right (zag), and returns the resulting leaf value:

zigzag :: Tree a → a
zigzag = zig

where
zig (Leaf x) = x
zig (Node l r) = zag l
zag (Leaf x) = x
zag (Node l r) = zag r

In a lazy language such as Haskell, evaluating zigzag (fullTree n) only builds as much of the
intermediate tree as necessary to produce the final result, which in this case is a single path.

However, due to the iterative nature of fullTree, in which the complete tree is potentially
traversed at each step in order to increase the depth by one, such an evaluation still requires
quadratic time, even in a lazy language. How can this be reduced to linear time?

6.1 The codensity monad

Voigtländer’s solution (2008) is based upon changing the representation of trees, using a
generalised form of continuation. Recall that a continuation can be viewed as a function
that is applied to the result of another computation. Using this idea, we can represent a
value x as the function λ c → c x that takes a continuation c, and applies this function to x
in order to produce the final result. This representation gives rise to the type (a → r) → r
of continuation computations of type a that return results of type r:

type Cont r a = (a → r) → r

It is easy to show that Cont r is a monadic type. Moreover, we can also parameterise the
declaration by another monad m to give a monad transformer (Liang et al., 1995):

type ContT r m a = (a → m r) → m r

For the purposes of improving the efficiency of fullTree, we will use the following gener-
alisation, known as the codensity monad transformer (Jaskelioff, 2009):

type CodT m a = ∀r. ((a → m r) → m r)

That is, the result type r is moved from the the left-side of the declaration to the right-side,
by exploiting Haskell’s notion of rank 2 types (Peyton Jones et al., 2007). Moving the
quantification in this manner means that whereas the continuation monad ContT r m has a
fixed result type r, the codensity monad CodT m has a variable (polymorphic) result type.
Making CodT into a monad transformer proceeds as follows:

instance Monad m ⇒ Monad (CodT m) where
return :: a → CodT m a
return x = λ c → c x

(>>=) :: CodT m a → (a → CodT m b) → CodT m b
f >>=g = λ c → f (λ x → g x c)

Using the codensity monad transformer, we now define a new representation for trees,
together with the necessary conversion functions between the original and new types:

type Coden a = CodT Tree a

rep :: Tree a → Coden a
rep t = (t >>=)

abs :: Coden a → Tree a
abs c = c return

It is interesting to note the similarity to the definitions rep xs = (xs++) and abs f = f [ ]
given earlier for lists. The above definitions for trees have the same structure, except that
the monoid operations ++ and [ ] are generalised to the monad operations >>= and return.
A simple calculation verifies the worker/wrapper assumption abs◦ rep = idTreea.

6.2 The term type

To improve the performance of fullTree, our aim now is to factorise this function into the
composition of a more efficient worker that produces a result in our codensity monad, and
a wrapper that converts the result back into the tree monad. That is, we seek to define a
function fullCoden that makes the following diagram commute:

Int
fullTree





fullCoden

��
Tree Int Coden Int

abs
��

Following the lead of our previous examples, we might expect to proceed by defining
fullTree as a fold over the type of natural numbers, and then applying our worker/wrapper
technique to derive the required worker. For this example, however, it turns out to be
preferable to begin by reformulating the problem in terms of a more structured type than
the natural numbers. Consider once again the definition for fullTree:

fullTree :: Int → Tree Int
fullTree 1 = return 1
fullTree (n +1) = fullTree n >>=λ i →

Node (return (n− i)) (return (i+1))

In this definition, the resulting trees are built using three functions:

return :: a → Tree a
(>>=) :: Tree a → (a → Tree b) → Tree b
Node :: Tree a → Tree a → Tree a

Based upon this observation, we can define the following type of tree terms whose values
represent trees that are built using these functions:

data Term a where
Return :: a → Term a
Bind :: Term a → (a → Term b) → Term b
Branch :: Term a → Term a → Term a

Reifying functions as data in this manner is sometimes called a deep embedding. Note that
because Bind involves terms of two different types, Term a is a GADT (Peyton Jones et al.,
2006). Categorically, defining a fold for such types requires moving to a functor category,
in which objects are functors and arrows are natural transformations (Johann & Ghani,

2008). In Haskell, the fold for terms can be defined as follows:

fold :: (∀a. a → f a) →
(∀ab. f a → (a → f b) → f b) →
(∀a. f a → f a → f a) →
(∀a. Term a → f a)

fold r b n (Return x) = r x
fold r b n (Bind t g) = b (fold r b n t) (fold r b n ◦g)
fold r b n (Branch t u) = n (fold r b n t) (fold r b n u)

The use of quantifiers in the type for fold reflects the use of natural transformations, which
in Haskell correspond to polymorphic functions. To ensure the expected universal property
we also require that Term and f are functors, but we omit the details here.

Using the fold operator for terms, the fact that terms represent trees can now be for-
malised by defining an evaluation function that simply replaces the syntactic constructors
on terms by the corresponding semantic operations on trees:

eval :: Term Int → Tree Int
eval = fold return (>>=) Node

In turn, we can define a version of fullTree that produces a term rather than a tree, by
replacing the use of the tree operations by the appropriate term constructors:

fullTerm :: Int → Term Int
fullTerm 1 = Return 1
fullTerm (n +1) = fullTerm n ‘Bind‘ λ i →

Branch (Return (n− i)) (Return (i+1))

A simple inductive proof shows that fullTree = eval◦ fullTerm.

6.3 Applying worker/wrapper

Having reformulated fullTree using an intermediate type of tree terms, we now seek to
complete the following expanded version of our commuting diagram from the previous
section, by defining appropriate functions work and fullCoden:

Int

fullTerm

��fullTree

��

fullCoden



Term Int

eval

����
��

��
��

��
�

work

���
��

��
��

���
�

Tree Int Coden Int
abs

��

Commutativity of the left triangle was established in the previous section. The following
definition ensures that the right triangle also commutes, by construction:

fullCoden :: Int → Coden Int
fullCoden = work ◦ fullTerm

In turn, we define the function work using fold for terms, by simply supplying the return
and >>= operations for our codensity monad, and a suitable node operation:

work :: Term Int → Coden Int
work = fold return (>>=) node

node :: Coden a → Coden a → Coden a
node f g = λ c → Node (f c) (g c)

To verify that this definition makes the lower triangle in the diagram commute, i.e. eval =
abs◦work, we begin by expanding out the definitions for eval and work to give:

fold return (>>=) Node = abs◦ fold return (>>=) node

Note that return and >>= on the left-side of the equation are for the tree monad, and on the
right-side are for the codensity monad. We then apply the worker/wrapper technique for
fold. In particular, condition (2) for this example expands to give three equations that are
together sufficient to justify the above factorisation:

(2.1) rep (return x) = return x

(2.2) rep (t >>= f ) = rep t >>= rep◦ f

(2.3) rep (Node l r) = node (rep l) (rep r)

The first two equations state that rep preserves the return and >>= operations and is hence
a monad morphism, while the last states that rep preserves the node operation. Verifying
these equations is simply a matter of expanding definitions and using monad laws. We
include all three proofs below to emphasise their simplicity.

Proof : (2.1)

rep (return x) g
= { applying rep }

return x >>=g
= { monad law }

g x
= { unapplying return for Coden }

return x g
��

Proof : (2.2)

rep (t >>= f ) g

= { applying rep }
(t >>= f )>>=g

= { monad law }
t >>=(λ x → f x >>=g)

= { unapplying rep }
t >>=(λ x → rep (f x) g)

= { unapplying rep }
rep t (λ x → rep (f x) g)

= { unapplying >>= for Coden }
(rep t >>= rep◦ f ) g

��
Proof : (2.3)

rep (Node l r) g
= { applying rep }

Node l r >>= g
= { applying >>= for Tree }

Node (l>>=g) (r >>= g)
= { unapplying rep }

Node (rep l g) (rep r g)
= { unapplying node }

node (rep l) (rep r) g
��

Finally, because the three internal triangles in the diagram commute, the external triangle
also commutes, which verifies the desired worker/wrapper factorisation:

fullTree :: Int → Tree Int
fullTree = abs◦ fullCoden

Returning to our original problem of improving the efficiency of zigzag (fullTree n),
if we now replace the original definition for fullTree by the new version obtained using
the worker/wrapper technique, the time complexity is reduced from quadratic to linear.
For example, in a simple experiment using the Glasgow Haskell Compiler the time for
n = 10000 was reduced from around 90 seconds to 0.2 seconds. If desired, the definition
fullCoden = work◦ fullTerm can also be fused to eliminate the use of the intermediate term
structure, further reducing the running time to under 0.1 seconds.

We conclude with a few remarks about this example. First of all, despite using a so-
phisticated optimisation technique in the form of the codensity monad, the proof of cor-
rectness of the efficient version of fullTerm still only requires simple equational reasoning.
Secondly, our proof of the worker/wrapper factorisation eval = work◦abs is not specific to
tree terms built using fullTerm, but shows how to optimise the evaluation of any such terms.
And finally, the use of tree terms also provides an explanation for why the optimisation is
correct, in the sense that it makes explicit the key idea of implementing the return, >>=,
and Node operations on expression trees using the codensity monad.

7 Conclusion and further work

In this article we developed a general worker/wrapper theory for changing the type of re-
cursive functions defined using fold operators, and showed how it can be used in practice as
an equational reasoning technique for improving the performance of functional programs.
The approach requires only basic categorical and equational reasoning principles, and
using fold operators results in simpler and more structured calculations than the previous
worker/wrapper theory based upon fixed point operators.

It is also interesting to recount how this work was developed. Initially we focused on
the special case of fold for lists, and identified conditions (1) and (2) for the case of lists.
However, it was not clear how these conditions were related, nor how they related to fold
fusion (3), or worker/wrapper fusion (6). It was only when we generalised from lists to an
arbitrary type using initial algebra semantics that it became clear that there were in fact four
relevant properties, related by a simple lattice structure. Focusing on lists made it difficult
to “see the wood for the trees”, and the move to a categorical approach revealed the simple
underlying algebraic structure of the problem.

There are many interesting topics for further work, including mechanising the tech-
nique, other recursion operators such as unfold, weaker versions of the worker/wrapper
assumption abs◦ rep = id, and other application areas. The monadic substitution example
also suggests a new approach to program optimisation that we are particularly keen to
explore, based upon a deep embedding of the operations to be optimised and the use of the
worker/wrapper technique to demonstrate correctness of the optimised program.

